

Counter-Flow Configuration

The operating conditions and geometry for the plate heat exchanger operating in a counter-flow configuration are shown schematically in Figure 8-36.




Figure 8-36: Plate heat exchanger in a counter-flow configuration.

The only difference between Figure 8-36 and the parallel-flow heat exchanger shown in Figure 8-32 is that the cold-side fluid enters at $x = L$ rather than $x = 0$. The state equations are obtained from the differential energy balances, also shown in Figure 8-36. The result is:

$$\frac{dT_H}{dx} = - \frac{2 N_{ch} (T_H - T_C)}{\dot{m}_H c_H \left(\frac{1}{h_H W} + \frac{th_m}{k_m W} + \frac{1}{h_C W} \right)} \quad (8-160)$$

$$\frac{dT_C}{dx} = - \frac{2 N_{ch} (T_H - T_C)}{\dot{m}_C c_C \left(\frac{1}{h_H W} + \frac{th_m}{k_m W} + \frac{1}{h_C W} \right)} \quad (8-161)$$

The inputs are entered in EES:

```
$UnitSystem SI MASS RAD PA K J
$Tabstops 0.2 0.4 0.6 3.5 in
```

"Inputs"

```
W=35 [cm]*convert(cm,m)
```

"width of heat exchanger"

```
L=70 [cm]*convert(cm,m)
```

"length of heat exchanger in flow direction"

```
N_ch=100 [-]
```

"number of channel pairs"

```
th_H=2.2 [mm]*convert(mm,m)
```

"channel width on hot-side"

```
th_C=2.2 [mm]*convert(mm,m)
```

"channel width on cold-side"

```
th_m=0.5 [mm]*convert(mm,m)
```

"thickness of plate"

E17: Section 8.6.2 Counter-Flow Configuration

```

p_H=7.5 [MPa]*convert(MPa,Pa)           "hot-side pressure"
p_C=100 [kPa]*convert(kPa,Pa)          "cold-side pressure"
m_dot_H=1.5 [kg/s]                      "hot-side mass flow rate"
m_dot_C=1.5 [kg/s]                      "cold-side mass flow rate"
H$='Air_ha'                             "hot-side fluid"
C$='Air_ha'                             "cold-side fluid"
T_H_in=300 [K]                          "hot-side inlet temperature"
T_C_in=90 [K]                           "cold-side inlet temperature"

```

A set of arbitrary state variables is used to establish that the state equations can be solved:

```

"arbitrary state variable to establish evaluation of the state equations"
x=0.1 [m]                               "position in heat exchanger"
T_H=300 [K]                              "hot-side temperature"
T_C=100 [K]                              "cold-side temperature"

```

The specific heat capacity and local heat transfer coefficient in each stream as well as the metal conductivity are determined:

```

c_H=cP(H$,p=p_H,T=T_H)                  "hot-side specific heat capacity"
c_C=cP(C$,p=p_C,T=T_C)                  "cold-side specific heat capacity"
call DuctFlow_local(H$,T_H,p_H,m_dot_H/N_ch,th_H,W,x,0 [-]:h_H, h_H_H , dpdx_H)
                                         "hot-side local heat transfer coefficient"
call DuctFlow_local(C$,T_C,p_C,m_dot_C/N_ch,th_C,W,x,0 [-]:h_C, h_H_C , dpdx_C)
                                         "cold-side local heat transfer coefficient"
k_m=k_('Aluminum', (T_H+T_C)/2)         "metal conductivity at local average temperature"

```

The state equations are evaluated according to Eqs. (8-161) and (8-162):

```

dTHdx=-2*N_ch*(T_H-T_C)/(m_dot_H*c_H*(1/(h_H*W)+th_m/(k_m*W)+1/(h_C*W)))
                                         "state equation for T_H"
dTcdx=-2*N_ch*(T_H-T_C)/(m_dot_C*c_C*(1/(h_H*W)+th_m/(k_m*W)+1/(h_C*W)))
                                         "state equation for T_C"

```

After establishing that the state equations can be successfully calculated, the arbitrary state variables are commented out:

```

{"arbitrary state variable to establish evaluation of the state equations"
x=0.1 [m]                               "position in heat exchanger"
T_H=300 [K]                              "hot-side temperature"
T_C=100 [K]                              "cold-side temperature"}

```

The state equations must be integrated from $x = 0$, where $T_H = T_{H,in}$ and $T_C = T_{C,out}$, to $x = L$, where $T_H = T_{H,out}$ and $T_C = T_{C,in}$. However, the value of $T_{C,out}$ is not known; therefore, a shooting technique is required. A value of $T_{C,out}$ is assumed:

```

T_C_out=290 [K]                          "assumed value of the cold outlet fluid"

```

The value of $T_{C,out}$ must be adjusted until the value of T_C predicted at $x = L$ is $T_{C,in}$. The integration is accomplished using the Integral command in EES:

E17: Section 8.6.2 Counter-Flow Configuration

N = 20 [-]	"number of integration steps"
DELTAx=L/N	"integration step size"
T_H=T_H_in+Integral(dTHdx,x,0,L,DELTAx)	"integral state equation for T_H"
T_C=T_C_out+Integral(dTCdx,x,0,L,DELTAx)	"integral state equation for T_C"

The iteration required by the shooting technique is accomplished using the optimization algorithm in EES. An objective function is defined as the absolute value of the difference between the calculated value of T_C at $x = L$ and the specified value of $T_{C,in}$.

err=Abs(T_C-T_C_in) "error between calculated and specified T_C_in"

The value of $T_{C,out}$ is commented out and Min/Max is selected from the Calculate menu. The value of the variable err is minimized by adjusting the variable T_C_out; a reasonable guess value (250 K) and bounds (220 K to 300 K) are specified. An integral table is created:

\$IntegralTable x:DELTAx,T_H,T_C

The optimization leads to $T_{C,out} = 221.1$ K and $\dot{q} = 200.0$ kW. The temperature distribution within the heat exchanger is illustrated in Figure 8-37.




Figure 8-37: Temperature distribution in the counter-flow heat exchanger.